

## Analysis of the Given Grammar

(February 20, 2024)

*COLORS in this document*

*Black: for original rules in the grammar in Language Specification document*

*Red: specifies needs for modifications*

*Green: rules do not require modification*

*Brown: Not valid syntactically*

The nonterminal  $\langle \text{program} \rangle$  is the start symbol of the given grammar.

1.  $\langle \text{program} \rangle \Rightarrow \langle \text{otherFunctions} \rangle \langle \text{mainFunction} \rangle$

The LL(1) property for the nonterminal  $\langle \text{program} \rangle$  is satisfied trivially due to single production for it. The following FOLLOW set will be required for verifying LL(1) compatibility for the nonterminal  $\langle \text{otherFunctions} \rangle$  in (3).

$\text{FOLLOW}(\langle \text{otherFunctions} \rangle) = \text{FIRST}(\langle \text{mainFunction} \rangle) = \{\text{TK\_MAIN}\}$

Note that all function definitions precede the main function definition and the language does not have constructs for function prototype declarations.

2.  $\langle \text{mainFunction} \rangle \Rightarrow \text{TK\_MAIN} \langle \text{stmts} \rangle \text{TK\_END}$

The LL(1) property for the nonterminal  $\langle \text{mainFunction} \rangle$  is satisfied trivially due to single production for it.

$\text{FIRST}(\langle \text{mainFunction} \rangle) = \text{FIRST}(\alpha) = \{\text{TK\_MAIN}\}$

where  $\alpha$  represents the right hand side of the production i.e.  $\text{TK\_MAIN} \langle \text{stmts} \rangle \text{TK\_END}$ .

3.  $\langle \text{otherFunctions} \rangle \Rightarrow \langle \text{function} \rangle \langle \text{otherFunctions} \rangle \mid \text{eps}$

The nonterminal  $\langle \text{otherFunctions} \rangle$  need special care for verifying the LL(1) compatibility due to a nullable production. Let us first verify that the sets  $\text{FIRST}(\langle \text{function} \rangle \langle \text{otherFunctions} \rangle)$  and  $\text{FOLLOW}(\langle \text{otherFunctions} \rangle)$  are disjoint.

$\text{FIRST}(\langle \text{function} \rangle \langle \text{otherFunctions} \rangle) = \text{FIRST}(\langle \text{function} \rangle) = \{\text{TK\_FUNID}\}$

Note that  $\langle \text{function} \rangle$  has no nullable production.

Also from 1, we have  $\text{FOLLOW}(\langle \text{otherFunctions} \rangle) = \{\text{TK\_MAIN}\}$

i.e.

$$\text{FIRST}(\langle \text{function} \rangle \langle \text{otherFunctions} \rangle) \cap \text{FOLLOW}(\langle \text{otherFunctions} \rangle) = \phi$$

The given productions for  $\langle \text{otherFunctions} \rangle$  are LL(1) compatible.

NOTE: Other properties such as ambiguity, left recursiveness, left factoring needs etc. causing violation of LL(1) compatibility of the rules will be discussed only if one or more of them exist. Otherwise I am focussing on violations due to epsilon productions. Also I will highlight the introduction of new nonterminals to incorporate the precedence of arithmetic operators and handling operations on variables of type record and union.

4.  $\langle \text{function} \rangle ::= \text{TK\_FUNID } \langle \text{input\_par} \rangle \langle \text{output\_par} \rangle \text{TK\_SEM } \langle \text{stmts} \rangle \text{TK\_END}$

This has no issue of violations in LL(1) compatibility. The nonterminal  $\langle \text{input\_par} \rangle$  is an essential construct to be part of the function definition while a function may or may not return values. Hence the  $\langle \text{output\_par} \rangle$  can have a syntax of (6) .

$$\text{FIRST}(\langle \text{function} \rangle) = \{\text{TK\_FUNID}\}$$

5.  $\langle \text{input\_par} \rangle ::= \text{TK\_INPUT TK\_PARAMETER TK\_LIST TK\_SQL } \langle \text{parameter\_list} \rangle \text{TK\_SQR}$

Single production having no conflict has its FIRST set as given follows:

$$\text{FIRST}(\langle \text{input\_par} \rangle) = \{\text{TK\_INPUT}\}$$

6.  $\langle \text{output\_par} \rangle ::= \text{TK\_OUTPUT TK\_PARAMETER TK\_LIST TK\_SQL } \langle \text{parameter\_list} \rangle \text{TK\_SQR} \mid \text{eps}$

Presence of epsilon production makes us verify whether  $\text{FIRST}(\alpha) \cap \text{FOLLOW}(\langle \text{output\_par} \rangle) = \phi$  or not, where  $\alpha$  represents the right hand side

$$\text{TK\_OUTPUT TK\_PARAMETER TK\_LIST TK\_SQL } \langle \text{parameter\_list} \rangle \text{TK\_SQR}$$

$$\text{FIRST}(\alpha) = \{\text{TK\_OUTPUT}\}$$

Refer rule 4 to compute the FOLLOW( $\langle \text{output\_par} \rangle$ ) as below

$$\text{FOLLOW}(\langle \text{output\_par} \rangle) = \{\text{TK\_SEM}\}$$

This implies that  $\text{FIRST}(\alpha) \cap \text{FOLLOW}(\langle \text{output\_par} \rangle) = \phi$

Hence the given rules for  $\langle \text{output\_par} \rangle$  conform to the LL(1) specifications.

7.  $\langle \text{parameter\_list} \rangle ::= \langle \text{dataType} \rangle \text{TK\_ID } \langle \text{remaining\_list} \rangle$

The single production for  $\langle \text{parameter\_list} \rangle$  is trivially LL(1) compatible.

$$\text{FIRST}(\langle \text{parameter\_list} \rangle) = \text{FIRST}(\langle \text{dataType} \rangle)$$

$$= \text{FIRST}(\langle \text{primitiveDatatype} \rangle) \cup \text{FIRST}(\langle \text{constructedDatatype} \rangle)$$

$$= \{\text{TK\_INT, TK\_REAL, TK\_RECORD, TK\_UNION}\}$$

Here we will allow the usage of record as well as union as parameters within the grammar as we did in this rule. This means that a function with following parameters will be considered as syntactically correct.

\_recordDemo1 input parameter list [record #book d5cc34, union #newbook d2cd, int d3, real d5]  
output parameter list[record #book d3];

But, a union variable passed as parameter or returned as a parameter will not make any sense due to the obvious reasons. However, this semantics will be captured in semantic analysis phase and you will have more clarity on type checking and semantic rules when your stage 2 work starts.

8.  $\langle \text{dataType} \rangle \implies \langle \text{primitiveDatatype} \rangle \mid \langle \text{constructedDatatype} \rangle$

$$\begin{aligned} \text{FIRST}(\langle \text{primitiveDatatype} \rangle) \cap \text{FIRST}(\langle \text{constructedDatatype} \rangle) \\ = \{ \text{TK\_INT}, \text{TK\_REAL} \} \cap \{ \text{TK\_RECORD}, \text{TK\_UNION} \} \\ = \phi \end{aligned}$$

Since there is no nullable production for  $\langle \text{dataType} \rangle$ , there is no need for computing FOLLOW( $\langle \text{dataType} \rangle$ ).

9.  $\langle \text{primitiveDatatype} \rangle \implies \text{TK\_INT} \mid \text{TK\_REAL}$

$$\text{FIRST}(\text{TK\_INT}) \cap \text{FIRST}(\text{TK\_REAL}) = \{ \text{TK\_INT} \} \cap \{ \text{TK\_REAL} \} = \phi$$

Hence the rules for  $\langle \text{primitiveDatatype} \rangle$  are LL(1) compatible.

$$\text{FIRST}(\langle \text{primitiveDatatype} \rangle) = \{ \text{TK\_INT}, \text{TK\_REAL} \}$$

10.  $\langle \text{constructedDatatype} \rangle \implies \text{TK\_RECORD TK\_RUID} \mid \text{TK\_UNION TK\_RUID}$

This non-terminal supports the parameter list or the declaration statements, which use the record and union datatype. If these are redefined using the  $\langle \text{defintype} \rangle$  or alias method as given below

```
defintype record #abc as #pqr
record #abc
    type int : x;
```

```
        type real:y;  
endrecord  
definetype record #definitionone as #definitiontwo  
record #definitionone  
    type int : z  
    type #pqr :a;          %non-recursive nested using definetype  
    type record #abc: b;    %non-recursive nested without using definetype  
    type #definitiontwo : c; %recursive nested using definetype  
endrecord
```

Consider the following declaration statements,

```
type record #definitionone c2d4;  
type #definitiontwo c5; %using the alias of redefinition as above.
```

Let us look into the requirement of modification in rules for <constructedDatatype> from the perspective of two usages of data type-one in the declaration statement and the other in the parameter list. Both can use record and union data types to declare type of their variables/ parameters. Also both can use the alias or the type definition. The newly defined type is represented by the token TK\_RUID.

Therefore the rules for <datatype> must include RUID in its first set also. The example in parameter list using both type names is as follows

```
_recordDemo1 input parameter list [ record #definitionone c2d4, #definitiontwo c5, record #abc d5cc34, #pqr d2cd]  
output parameter list[record #abc d3];
```

Therefore, modify the rules for <constructedDatatype> as follows

$\langle \text{constructedDatatype} \rangle ::= \text{TK\_RECORD TK\_RUID} \mid \text{TK\_UNION TK\_RUID} \mid \text{TK\_RUID}$  .....10-a

$\text{FIRST}(\langle \text{constructedDatatype} \rangle) = \{\text{TK\_RECORD}, \text{TK\_UNION}, \text{TK\_RUID}\}$

Also since the first sets of RHS of the three rules for  $\langle \text{constructedDatatype} \rangle$  are disjoint, the grammar is LL(1).

That is  $\text{FIRST}(\text{TK\_RECORD TK\_RUID}) \cap \text{FIRST}(\text{TK\_UNION TK\_RUID}) \cap \text{FIRST}(\text{TK\_RUID})$

$= \{\text{TK\_RECORD}\} \cap \{\text{TK\_UNION}\} \cap \{\text{TK\_RUID}\}$

$= \phi$

11.  $\langle \text{remaining\_list} \rangle ::= \text{TK\_COMMA} \langle \text{parameter\_list} \rangle \mid \text{eps}$

There exist a nullable production for the non-terminal  $\langle \text{remaining\_list} \rangle$ . The RHS (right hand side) of the other production is such that the null string eps (epsilon) is not derivable from it.

Also

$\text{FIRST}(\text{TK\_COMMA} \langle \text{parameter\_list} \rangle) \cap \text{FOLLOW}(\langle \text{remaining\_list} \rangle)$  should be empty.

$\text{FIRST}(\text{TK\_COMMA} \langle \text{parameter\_list} \rangle) = \{\text{TK\_COMMA}\}$

and

$\text{FOLLOW}(\langle \text{remaining\_list} \rangle) = \text{FOLLOW}(\text{parameter\_list})$ ..... from (7)

$= \{\text{TK\_SQR}\}$

Hence,

$\text{FIRST}(\text{TK\_COMMA} \langle \text{parameter\_list} \rangle) \cap \text{FOLLOW}(\langle \text{remaining\_list} \rangle)$

$= \{\text{TK\_COMMA}\} \cap \{\text{TK\_SQR}\}$

$= \phi$

Therefore both the given rules for the non-terminal  $\langle \text{remaining\_list} \rangle$  conform to LL(1) specifications.

12.  $\langle \text{stmts} \rangle ::= \langle \text{typeDefinitions} \rangle \langle \text{declarations} \rangle \langle \text{otherStmts} \rangle \langle \text{returnStmt} \rangle$

The nonterminal  $\langle \text{stmts} \rangle$  specifies the grammar for the body of the function. The ordering of other nonterminals such as  $\langle \text{typeDefinitions} \rangle$ ,  $\langle \text{declarations} \rangle$  and  $\langle \text{returnStmt} \rangle$  have fixed positions within the body of the function code.

There is no epsilon production  $\langle \text{in} \langle \text{stmts} \rangle$  and a single non nullable production for  $\langle \text{stmts} \rangle$  is LL(1) compatible.

In order to find the first set of  $\langle \text{stmts} \rangle$ , we need to union the first sets as follows. The non-terminals  $\langle \text{typeDefinitions} \rangle$ ,  $\langle \text{declarations} \rangle$  and  $\langle \text{otherStmts} \rangle$  are nullable and have reachability through the application of their null productions.

$$\text{FIRST}(\langle \text{stmts} \rangle) = \text{FIRST}(\langle \text{typeDefinitions} \rangle) \cup \text{FIRST}(\langle \text{declarations} \rangle) \cup \text{FIRST}(\langle \text{otherStmts} \rangle) \cup \text{FIRST}(\langle \text{returnStmt} \rangle)$$

.....using 13, 19, 22 and 43

$$\begin{aligned} &= \{\text{TK\_RECORD}, \text{TK\_UNION}\} \cup \{\text{TK\_TYPE}\} \cup \{\text{TK\_ID}, \text{TK\_RUID}, \text{TK\_WHILE}, \text{TK\_IF}, \text{TK\_READ}, \text{TK\_WRITE}, \text{TK\_SQL}, \\ &\text{TK\_CALL}\} \cup \{\text{TK\_RETURN}\} \\ &= \{\text{TK\_RECORD}, \text{TK\_UNION}, \text{TK\_TYPE}, \text{TK\_ID}, \text{TK\_RUID}, \text{TK\_WHILE}, \text{TK\_IF}, \text{TK\_READ}, \text{TK\_WRITE}, \text{TK\_SQL}, \\ &\text{TK\_CALL}, \text{TK\_RETURN}\} \end{aligned}$$

Since the nonterminal  $\langle \text{stmts} \rangle$  does not have a nullable production, we need not compute FOLLOW( $\langle \text{stmts} \rangle$ ) for populating the parsing table. Recall the construction of the parsing table. The columns corresponding to the tokens in FIRST( $\langle \text{stmts} \rangle$ ) corresponding to the row  $\langle \text{stmts} \rangle$  are populated with the rule number 12.

13.  $\langle \text{typeDefinitions} \rangle ::= \langle \text{typeDefinition} \rangle \langle \text{typeDefinitions} \rangle \mid \epsilon$

This is the place where the  $\langle \text{definetypstmt} \rangle$  finds its existence. As it is used for record and union types redefinitions (or alias). Let us have this construct appear anywhere within the piece of code reserved for  $\langle \text{typeDefinitions} \rangle$

Redefine the rule 13

$$\langle \text{typeDefinitions} \rangle ::= \langle \text{actualOrRedefined} \rangle \langle \text{typeDefinitions} \rangle \mid \epsilon \quad \text{....13-a}$$

Where

$$\langle \text{actualOrRedefined} \rangle ::= \langle \text{typeDefinition} \rangle \mid \langle \text{definetypstmt} \rangle \quad \text{.....13-b}$$

With this, the constructs  $\langle \text{typeDefinition} \rangle$  and  $\langle \text{definetypstmt} \rangle$  can appear in any order before the  $\langle \text{stmts} \rangle$  construct within a function.

The semantic analyzer will be able to catch the relevance of the presence of a type redefinition and type equivalence later. If your test case has the following code

```

definetype union #old as #new           % line 1- derived using <definetypestmt>.
% following record type definition is derived using <typeDefinition>
record #definitionone
    type int : z;
    type #pqr :a;           %non-recursive nested using definetype
    type record #abc: b;    %non-recursive nested without using definetype
    type #definitiontwo : c; % line 2
endrecord
definetype record #definitionone as #definitiontwo % derived using <definetypestmt>.

```

Note that here the record #old in line 1 is not defined using <typeDefinition> grammar but will still be syntactically correct. The semantic analyzer will be able to catch this error. Also note that the record #definitionone is redefined as #definitiontwo even after its use above in line - 2. The semantic analyzer will require two pass mechanism to collect all information.

Let us analyze 13-b first.

Since  $\text{FIRST}(\langle \text{typeDefinition} \rangle) \cap \text{FIRST}(\langle \text{definetypestmt} \rangle)$

$= \{\text{TK\_RECORD}, \text{TK\_UNION}\} \cap \{\text{TK\_DEFINETYPE}\} \dots\dots\dots$  using 14, 15 and 46

Therefore rule 13-b is LL(1) compatible.

And  $\text{FIRST}(\langle \text{actualOrRedefined} \rangle) = \{\text{TK\_RECORD}, \text{TK\_UNION}, \text{TK\_DEFINETYPE}\}$

Now 13-a

$\text{FIRST}(\langle \text{actualOrRedefined} \rangle) \cap \text{FOLLOW}(\langle \text{typeDefinitions} \rangle)$

$= \{\text{TK\_RECORD}, \text{TK\_UNION}, \text{TK\_DEFINETYPE}\} \cap \text{FIRST}(\langle \text{declarations} \rangle)$

$$\begin{aligned}
&= \{TK\_RECORD, TK\_UNION, TK\_DEFINETYPE\} \cap FIRST(<declaration>) \\
&= \{TK\_RECORD, TK\_UNION, TK\_DEFINETYPE\} \cap \{TK\_TYPE\} \\
&= \phi
\end{aligned}$$

Therefore, rule 13-a is LL(1) compatible

14.  $\langle typeDefinition \rangle \Rightarrow TK\_RECORD \ TK\_RUID \ \langle fieldDefinitions \rangle \ TK\_ENDRECORD$

15.  $\langle typeDefinition \rangle \Rightarrow TK\_UNION \ TK\_RUID \ \langle fieldDefinitions \rangle \ TK\_ENDUNION$

The rule 14 reaches to the

$FIRST(TK\_RECORD \ TK\_RUID \ \langle fieldDefinitions \rangle \ TK\_ENDRECORD) = \{TK\_RECORD\}$

Similarly, rule 15 reaches the first of its RHS

$FIRST(TK\_UNION \ TK\_RUID \ \langle fieldDefinitions \rangle \ TK\_ENDUNION) = \{TK\_UNION\}$

Also, the Firsts sets of the RHS of the two rules 14 and 15 for the non-terminal  $\langle typeDefinition \rangle$  are disjoint.

$$\{TK\_RECORD\} \cup \{TK\_UNION\} = \phi$$

Therefore rules 14 and 15 conform to the LL(1) specifications and do not require any modification.

Also,  $FIRST(\langle typeDefinition \rangle) = \{TK\_RECORD, TK\_UNION\}$

16.  $\langle fieldDefinitions \rangle \Rightarrow \langle fieldDefinition \rangle \langle fieldDefinition \rangle \langle moreFields \rangle$

As two consecutive occurrences of  $\langle fieldDefinition \rangle$  are there at the RHS of the above production, this imposes a requirement of at least two fields in a record. There can be one or more fields afterwards. As the above rule is single and non nullable, it is LL(1) compatible.

$$FIRST(\langle fieldDefinitions \rangle) = FIRST(\langle fieldDefinition \rangle) = \{TK\_TYPE\}$$

17.  $\langle fieldDefinition \rangle \Rightarrow TK\_TYPE \ \langle primitiveDatatype \rangle \ TK\_COLON \ TK\_FIELDID \ TK\_SEM$

Here, the type of the field needs major change to support the nested record and union data types. As per the descriptions given in the language specification document, a field name can also be of type union or record. However, only the alias of the union or record type is allowed in the type



definition of these fields. A minor correction in the example given in the language specifications document that uses type newname: field should be updated using a # as prefix. The updated example is given below.

```
definetype union #student as #newname
record #taggedunionexample
    type int : tagvalue;
    type #newname : field; %add # before newname which was missing in the document earlier.
endrecord
```

Therefore the rule for <fielddefinition> is modified as given below.

<fieldDefinition>====> TK\_TYPE <fieldType> TK\_COLON TK\_FIELDID TK\_SEM

<fieldtype>====> <primitiveDatatype> | TK\_RUID

..... Add this rule in the grammar 17(a)

Note that the record or union data type can be made nested by using the alias. This grammar will not support the usage of usual type definition syntax of union and record to be nested here. To explain this, the following example **will not be syntactically valid**.

```
record #taggedunionexample
    type int : tagvalue;
    type union #student
        type int:rollno;
        type real marks;
        type int age;
    endunion : field; %add # before newname which was missing in the document earlier.
endunion
endrecord
```

%NOT CORRECT SYNTACTICALLY

Now check the compatibility of the two rules in 17(a)

$\text{FIRST}(\langle \text{primitiveDatatype} \rangle) \cap \text{FIRST}(\text{TK\_RUID}) = \{\text{TK\_INT}, \text{TK\_REAL}\} \cap \{\text{TK\_RUID}\} = \phi$

And  $\text{FIRST}(\langle \text{fieldType} \rangle) = \{\text{TK\_INT}, \text{TK\_REAL}, \text{TK\_RUID}\}$

$\text{FIRST}(\langle \text{fieldDefinition} \rangle) = \{\text{TK\_TYPE}\}$

18.  $\langle \text{moreFields} \rangle \Rightarrow \langle \text{fieldDefinition} \rangle \langle \text{moreFields} \rangle \mid \epsilon$

Let us find

$\text{FOLLOW}(\langle \text{moreFields} \rangle) = \phi \cup \text{FOLLOW}(\langle \text{fieldDefinitions} \rangle) = \{\text{TK\_ENDRECORD}, \text{TK\_ENDUNION}\}$  ..... using rules 14, 15 and 16

$$\begin{aligned} & \text{FIRST}(\langle \text{fieldDefinition} \rangle \langle \text{moreFields} \rangle) \cap \text{FOLLOW}(\langle \text{moreFields} \rangle) \\ &= \{\text{TK\_TYPE}\} \cap \{\text{TK\_ENDRECORD}, \text{TK\_ENDUNION}\} \\ &= \phi \end{aligned}$$

Hence, the above two rules for the non terminal  $\langle \text{moreFields} \rangle$  are LL(1) compatible.

19.  $\langle \text{declarations} \rangle \Rightarrow \langle \text{declaration} \rangle \langle \text{declarations} \rangle \mid \epsilon$

$$\begin{aligned} \text{FOLLOW}(\langle \text{declarations} \rangle) &= \text{FIRST}(\langle \text{otherStmts} \rangle) \cup \text{FOLLOW}(\langle \text{otherStmts} \rangle) \\ &= \{\text{TK\_ID}, \text{TK\_RUID}, \text{TK\_WHILE}, \text{TK\_IF}, \text{TK\_READ}, \text{TK\_WRITE}, \text{TK\_SQL}, \text{TK\_CALL}\} \end{aligned}$$

$$\text{FIRST}(\langle \text{declaration} \rangle \langle \text{declarations} \rangle) = \text{FIRST}(\langle \text{declaration} \rangle) = \{\text{TK\_TYPE}\}$$

We observe that both of the above sets are disjoint. Therefore both the above productions for the non terminal  $\langle \text{declarations} \rangle$  conform to LL(1) specifications.

20.  $\langle \text{declaration} \rangle \Rightarrow \text{TK\_TYPE} \langle \text{dataType} \rangle \text{TK\_COLON} \text{TK\_ID} \text{TK\_COLON} \langle \text{global\_or\_not} \rangle \text{TK\_SEM}$

$\langle \text{declaration} \rangle \Rightarrow \text{TK\_TYPE} \langle \text{dataType} \rangle \text{TK\_COLON} \text{TK\_ID} \langle \text{global\_or\_not} \rangle \text{TK\_SEM}$

Also to incorporate the optional existence of global keyword prefixed with a colon is reflected as modification in rule 21.

$\text{FIRST}(\langle \text{declaration} \rangle) = \{\text{TK\_TYPE}\}$

Rule conforms to LL(1) specifications.

21.  $\langle \text{global\_or\_not} \rangle \Rightarrow \text{TK\_GLOBAL} \mid \epsilon$

This rule is also modified as follows

$\langle \text{global\_or\_not} \rangle \Rightarrow \text{TK\_COLON TK\_GLOBAL} \mid \text{eps}$

$\text{FOLLOW}(\langle \text{global\_or\_not} \rangle) = \{\text{TK\_SEM}\}$  ..... using 20

and

$\text{FIRST}(\text{TK\_COLON TK\_GLOBAL}) = \{\text{TK\_COLON}\}$

As

$\text{FOLLOW}(\langle \text{global\_or\_not} \rangle) \cap \text{FIRST}(\text{TK\_COLON TK\_GLOBAL}) = \{\text{TK\_SEM}\} \cap \{\text{TK\_COLON}\} = \phi$

Therefore the new rule for  $\langle \text{global\_or\_not} \rangle$  is LL(1) compatible.

22.  $\langle \text{otherStmts} \rangle \Rightarrow \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \mid \text{eps}$

$\text{FOLLOW}(\langle \text{otherStmts} \rangle) = \text{FIRST}(\langle \text{returnStmts} \rangle)$

$\text{FIRST}(\langle \text{otherStmts} \rangle) = \text{FIRST}(\langle \text{stmt} \rangle)$  [ Note:  $\langle \text{stmt} \rangle$  does not derive an empty string]  
 $= \{\text{TK\_ID}, \text{TK\_WHILE}, \text{TK\_IF}, \text{TK\_READ}, \text{TK\_WRITE}, \text{TK\_SQL}, \text{TK\_CALL}\}$

.....using 23

23.  $\langle \text{stmt} \rangle \Rightarrow \langle \text{assignmentStmt} \rangle \mid \langle \text{iterativeStmt} \rangle \mid \langle \text{conditionalStmt} \rangle \mid \langle \text{ioStmt} \rangle \mid \langle \text{funCallStmt} \rangle$

Referring rules 24, 29, 30, 32 and 26, we conclude that the five productions given above for the nonterminal  $\langle \text{stmt} \rangle$  are not nullable.

We must check the LL(1) property that the FIRST sets of the RHSs of all productions above are disjoint.

$\text{FIRST}(\langle \text{assignmentStmt} \rangle) \cap \text{FIRST}(\langle \text{iterativeStmt} \rangle) \cap \text{FIRST}(\langle \text{conditionalStmt} \rangle) \cap \text{FIRST}(\langle \text{ioStmt} \rangle) \cap \text{FIRST}(\langle \text{funCallStmt} \rangle)$   
 $= \{\text{TK\_ID}\} \cap \{\text{TK\_WHILE}\} \cap \{\text{TK\_IF}\} \cap \{\text{TK\_READ}, \text{TK\_WRITE}\} \cap \{\text{TK\_SQL}, \text{TK\_CALL}\}$   
 $= \phi$

Hence, the above five productions for the nonterminal  $\langle \text{stmt} \rangle$  conform to LL(1) specifications.

and ,

$\text{FIRST}(\langle \text{stmt} \rangle) = \{\text{TK\_ID}, \text{TK\_WHILE}, \text{TK\_IF}, \text{TK\_READ}, \text{TK\_WRITE}, \text{TK\_SQL}, \text{TK\_CALL}\}$

24.  $\langle \text{assignmentStmt} \rangle \Rightarrow \langle \text{SingleOrRecId} \rangle \text{TK\_ASSIGNOP} \langle \text{arithmeticExpression} \rangle \text{TK\_SEM}$

$\text{FIRST}(\langle \text{assignmentStmt} \rangle) = \text{FIRST}(\langle \text{singleOrRecId} \rangle) = \{\text{TK\_ID}\}$  ..... using 25 e

25.  $\langle \text{singleOrRecId} \rangle ::= \text{TK\_ID} \mid \text{TK\_ID TK\_DOT TK\_FIELDID}$ 

The second rule for  $\langle \text{singleOrRecId} \rangle$  does not support the record or union type variable construction expanded using the dot operator for any number of times. As you have seen in the examples given under rules for  $\langle \text{constructedDatatype} \rangle$ , if a variable is declared as follows

```
definetype record #definitionone as #definitiontwo
  record #definitionone
    type int : z
    type #pqr : a;           %non-recursive nested using definetype
    type record #abc: b;     %non-recursive nested without using definetype
    type #definitiontwo : c; %recursive nested using definetype
  endrecord
```

Consider the following declaration statements,

```
type record #definitionone c2d4;
type #definitiontwo c5;
```

and different variables constructed as c2d4.c, c5.b, c2cd4.c.c.z, c2cd4.c.c.c.c.c.z, and so on.

Then the following rules for  $\langle \text{singleOrRecId} \rangle$  will not help as the token stream would then be TK\_ID TK\_DOT TK\_FIELDID TK\_DOT TK\_FIELDID TK\_DOT TK\_FIELDID TK\_DOT TK\_FIELDID TK\_DOT TK\_FIELDID TK\_DOT TK\_FIELDID TK\_DOT TK\_FIELDID for a variable c2cd4.c.c.c.c.c.z (for example)

$\langle \text{singleOrRecId} \rangle ::= \text{TK\_ID} \mid \text{TK\_ID TK\_DOT TK\_FIELDID}$

So we need to redesign the rule

$\langle \text{singleOrRecId} \rangle ::= \text{TK\_ID TK\_DOT TK\_FIELDID}$

As

$\langle \text{singleOrRecId} \rangle ::= \langle \text{constructedVariable} \rangle$

by introducing a new nonterminal say  $\langle \text{constructedVariable} \rangle$  which will represent the variables as described above.

Therefore the rules for  $\langle \text{singleOrRecId} \rangle$  will become

$\langle \text{singleOrRecId} \rangle ::= \text{TK\_ID} \mid \langle \text{constructedVariable} \rangle$  ..... 25 a

We can define a recursive rule for this as follows.

$\langle \text{constructedVariable} \rangle ::= \text{TK\_ID} \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle$  ----- 25 b

$\langle \text{oneExpansion} \rangle ::= \text{TK\_DOT TK\_FIELDID}$  ----- 25 c

$\langle \text{moreExpansions} \rangle ::= \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle \mid \epsilon$  ----- 25 d

Here two more non-terminals are introduced to address nested record types for the variables constructed using them. It is clear that the nesting level of zero, i.e. with no record type field, the nonterminal  $\langle \text{oneExpansion} \rangle$  derives  $\text{TK\_DOT TK\_FIELDID}$  and in this case the null production for  $\langle \text{moreExpansions} \rangle$  is used. For nested definitions, a number of times  $\langle \text{moreExpansions} \rangle ::= \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle$  rule is applied as required..

If you wish you can replace the name  $\langle \text{singleOrRecId} \rangle$  with  $\langle \text{singleOrRecOrUnionID} \rangle$  everywhere to make it self-explanatory.

Before going for verifying the LL(1) compatibility of the rule for  $\langle \text{singleOrRecId} \rangle$ , let us examine the newly introduced rules.

Rule 25 b is a single rule and is non nullable.

$\text{FIRST}(\langle \text{constructedVariable} \rangle) = \{\text{TK\_ID}\}$

$\text{FOLLOW}(\langle \text{constructedVariable} \rangle) = \text{FOLLOW}(\langle \text{singleOrRecId} \rangle) = \{\text{TK\_ASSIGNOP}\}$

Rule 25 c is LL(1) trivially.

$$\text{FIRST}(\langle \text{oneExpansion} \rangle) = \{\text{TK\_DOT}\}$$

Rule 25 d requires that  $\text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle)$  and  $\text{FOLLOW}(\langle \text{moreExpansions} \rangle)$  should be disjoint.

$$\begin{aligned} & \text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle) \cap \text{FOLLOW}(\langle \text{moreExpansions} \rangle) \\ &= \{\text{TK\_DOT}\} \cap \text{FOLLOW}(\langle \text{constructedVariable} \rangle) \\ &= \{\text{TK\_DOT}\} \cap \text{FOLLOW}(\langle \text{singleOrRecId} \rangle) \\ &= \{\text{TK\_DOT}\} \cap \{\text{TK\_ASSIGNOP}, \text{TK\_CL}\} \\ &= \emptyset \end{aligned}$$

Using 25 a and 25 b

$$\langle \text{singleOrRecId} \rangle ::= \text{TK\_ID} \mid \text{TK\_ID} \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle$$

Both of the above productions for the nonterminal  $\langle \text{singleOrRecId} \rangle$  are non nullable and their first sets are not disjoint. This requires left factoring

The above two rules are left factored using new non terminal  $\langle \text{new\_24} \rangle$

$$\langle \text{singleOrRecId} \rangle ::= \text{TK\_ID} \langle \text{option\_single\_constructed} \rangle \quad \dots\dots\dots 25 \text{ e}$$

$$\langle \text{option\_single\_constructed} \rangle ::= \epsilon \mid \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle \quad \dots\dots\dots 25 \text{ f}$$

25 e is trivially LL(1) compatible.

For both rules for  $\langle \text{option\_single\_constructed} \rangle$  in 25 f,  $\text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle)$  and  $\text{FOLLOW}(\langle \text{option\_single\_constructed} \rangle)$  should be disjoint.

$$\begin{aligned} & \text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle) \cap \text{FOLLOW}(\langle \text{option\_single\_constructed} \rangle) \\ &= \{\text{TK\_DOT}\} \cap \text{FOLLOW}(\langle \text{singleOrRecId} \rangle) \\ &= \{\text{TK\_DOT}\} \cap \{\text{TK\_ASSIGNOP}, \text{TK\_CL}\} \end{aligned}$$

$= \phi$ .

Therefore, the above productions are LL(1) compatible.

The final rules for  $\langle \text{singleOrRecId} \rangle$  include 25 e, f, c and d.

Using 25 e,  $\text{FIRST}(\langle \text{singleOrRecId} \rangle) = \{\text{TK\_ID}\}$

26.  $\langle \text{funCallStmt} \rangle \Rightarrow \langle \text{outputParameters} \rangle \text{ TK\_CALL TK\_FUNID TK\_WITH TK\_PARAMETERS } \langle \text{inputParameters} \rangle$

The rule must have at its end TK\_SEM to represent semicolon

$\langle \text{funCallStmt} \rangle \Rightarrow \langle \text{outputParameters} \rangle \text{ TK\_CALL TK\_FUNID TK\_WITH TK\_PARAMETERS } \langle \text{inputParameters} \rangle \text{ TK\_SEM}$

The rule is trivially LL(1) compatible.

$\text{FIRST}(\langle \text{funCallStmt} \rangle) = \text{FIRST}(\langle \text{outputParameters} \rangle) \cup \text{FOLLOW}(\langle \text{outputParameters} \rangle)$   
 $= \{\text{TK\_SQL}\} \cup \{\text{TK\_CALL}\} = \{\text{TK\_SQL}, \text{TK\_CALL}\}$

27.  $\langle \text{outputParameters} \rangle \Rightarrow \text{TK\_SQL } \langle \text{idList} \rangle \text{ TK\_SQR TK\_ASSIGNOP } | \in$

$\text{FOLLOW}(\langle \text{outputParameters} \rangle) = \{\text{TK\_CALL}\}$

Due to the presence of the nullable production for the non terminal  $\langle \text{outputParameters} \rangle$ , we check the following LL(1) property

$\text{FIRST}(\text{TK\_SQL } \langle \text{idList} \rangle \text{ TK\_SQR TK\_ASSIGNOP}) \cap \text{FOLLOW}(\langle \text{outputParameters} \rangle)$   
 $= \{\text{TK\_SQL}\} \cap \{\text{TK\_CALL}\} = \phi$

Hence, the productions in 27 are LL(1) compatible.

$\text{FIRST}(\langle \text{outputParameters} \rangle) = \{\text{TK\_SQL}\}$

28.  $\langle \text{inputParameters} \rangle \Rightarrow \text{TK\_SQL } \langle \text{idList} \rangle \text{ TK\_SQR}$

LL(1) compatible with  $\text{FIRST}(\langle \text{inputParameters} \rangle) = \{\text{TK\_SQL}\}$

29.  $\langle \text{iterativeStmt} \rangle \Rightarrow \text{TK\_WHILE TK\_OP } \langle \text{booleanExpression} \rangle \text{ TK\_CL } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK\_ENDWHILE}$

LL(1) compatible with  $\text{FIRST}(\langle \text{iterativeStmt} \rangle) = \{\text{TK\_WHILE}\}$

30.  $\langle \text{conditionalStmt} \rangle \Rightarrow \text{TK\_IF } \langle \text{booleanExpression} \rangle \text{ TK\_THEN } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK\_ELSE } \langle \text{otherStmts} \rangle \text{ TK\_ENDIF}$

31.  $\langle \text{conditionalStmt} \rangle \Rightarrow \text{TK\_IF } \langle \text{booleanExpression} \rangle \text{ TK\_THEN } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK\_ENDIF}$

The two rules (30) and (31) for the nonterminal  $\langle \text{conditionalStmt} \rangle$  need modifications as they originally form ambiguous grammar. We need to perform the left factoring as follows.

$\langle \text{conditionalStmt} \rangle \Rightarrow \text{TK\_IF } \text{TK\_OP } \langle \text{booleanExpression} \rangle \text{ TK\_CL TK\_THEN } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \langle \text{elsePart} \rangle \dots \dots \dots$  number rule as 30-a  
 $\langle \text{elsePart} \rangle \Rightarrow \text{TK\_ELSE } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK\_ENDIF} \mid \text{TK\_ENDIF} \dots \dots \dots$  number rules as 31-a

Notice that I also added TK\_OP and TK\_CL in new rule 30-a to enclose boolean expression. These two tokens were missing earlier.

Also I am putting a constraint of at least one statement within else part.

Now there is a single rule for  $\langle \text{conditionalStmt} \rangle$  which is non nullable and its first set is  $\{\text{TK\_IF}\}$

But the newly introduced nonterminal  $\langle \text{elsePart} \rangle$  has two non nullable productions and their corresponding FIRST sets are disjoint.

$\text{FIRST}(\text{TK\_ELSE } \langle \text{otherStmts} \rangle \text{ TK\_ENDIF}) \cap \text{FIRST}(\text{TK\_ENDIF}) = \{\text{TK\_ELSE}\} \cap \{\text{TK\_ENDIF}\} = \phi$

Therefore the new set of rules for  $\langle \text{elsePart} \rangle$  is also LL(1) compatible.

And,

$\text{FIRST}(\langle \text{conditionalStmt} \rangle) = \{\text{TK\_IF}\}$

$\text{FIRST}(\langle \text{elsePart} \rangle) = \{\text{TK\_ELSE}, \text{TK\_ENDIF}\}$

32.  $\langle \text{ioStmt} \rangle \Rightarrow \text{TK\_READ TK\_OP } \langle \text{var} \rangle \text{ TK\_CL TK\_SEM} \mid \text{TK\_WRITE TK\_OP } \langle \text{var} \rangle \text{ TK\_CL TK\_SEM}$

Both the above productions are non nullable.

and,

$\text{FIRST}(\text{TK\_READ TK\_OP } \langle \text{var} \rangle \text{ TK\_CL TK\_SEM}) \cap \text{FIRST}(\text{TK\_WRITE TK\_OP } \langle \text{var} \rangle \text{ TK\_CL TK\_SEM})$



$$= \{TK\_READ\} \cap \{TK\_WRITE\} = \phi$$

Therefore, the above productions are LL(1) compatible.

and,

$$FIRST(<ioStmt>) = \{TK\_READ, TK\_WRITE\}$$

33.  $\langle arithmeticExpression \rangle \implies \langle arithmeticExpression \rangle \langle operator \rangle \langle arithmeticExpression \rangle$

I had left a major correction for you to do in the arithmetic expression grammar. You were required to impose precedence of operators and the operations on record variables.

34.  $\langle arithmeticExpression \rangle \implies TK\_OP \langle arithmeticExpression \rangle TK\_CL \mid \langle var \rangle$

An arithmetic expression is expected to take as argument an identifier, statically available integer number, real number, a record identifier and a record variable identifier (constructed by expanding the name using dot with a field name). These are represented in the non-terminal  $\langle var \rangle$  in rule number 39-a.

Now we reformulate the rules 33, 34 and 35 to impose precedence.

$\langle arithmeticExpression \rangle \implies \langle arithmeticExpression \rangle \langle lowPrecedenceOperators \rangle \langle term \rangle \mid \langle term \rangle \dots\dots\dots (A1)$

$\langle term \rangle \implies \langle term \rangle \langle highPrecedenceOperators \rangle \langle factor \rangle \mid \langle factor \rangle \dots\dots\dots (A2)$

$\langle factor \rangle \implies TK\_OP \langle arithmeticExpression \rangle TK\_CL \mid \langle var \rangle \dots\dots\dots (A3)$

$\langle highPrecedenceOperator \rangle \implies TK\_MUL \mid TK\_DIV \dots\dots\dots (A4)$

$\langle lowPrecedenceOperators \rangle \implies TK\_PLUS \mid TK\_MINUS \dots\dots\dots (A5)$

[Note : I have given new numbers for referring to the rules, but you can have your own numbering scheme for the rules]

Now let us verify the rules A1-5 for the nonterminals  $\langle arithmeticExpression \rangle$ ,  $\langle term \rangle$ ,  $\langle factor \rangle$ ,  $\langle lowPrecedenceOperators \rangle$ ,

$\langle highPrecedenceOperators \rangle$ ,

The rule A1 needs left recursion elimination and the new set of productions are obtained by introducing new nonterminal, say  $\langle expPrime \rangle$

$\langle arithmeticExpression \rangle \implies \langle term \rangle \langle expPrime \rangle$

$\langle expPrime \rangle \implies \langle lowPrecedenceOperators \rangle \langle term \rangle \langle expPrime \rangle \mid \epsilon$

Similarly rule A2 also needs left recursion elimination. The new set of productions are obtained by introducing new nonterminal, say,  $\langle \text{termPrime} \rangle$

$\langle \text{term} \rangle \Rightarrow \langle \text{factor} \rangle \langle \text{termPrime} \rangle$

$\langle \text{termPrime} \rangle \Rightarrow \langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle \mid \epsilon$

Now we have two more nonterminals  $\langle \text{expPrime} \rangle$  and  $\langle \text{termPrime} \rangle$ . The complete set of non left recursive, unambiguous arithmetic expression grammar then becomes with new numbers B1-B7

B1)  $\langle \text{arithmeticExpression} \rangle \Rightarrow \langle \text{term} \rangle \langle \text{expPrime} \rangle$

B2)  $\langle \text{expPrime} \rangle \Rightarrow \langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle \mid \epsilon$

B3)  $\langle \text{term} \rangle \Rightarrow \langle \text{factor} \rangle \langle \text{termPrime} \rangle$

B4)  $\langle \text{termPrime} \rangle \Rightarrow \langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle \mid \epsilon$

B5)  $\langle \text{factor} \rangle \Rightarrow \text{TK\_OP} \langle \text{arithmeticExpression} \rangle \text{TK\_CL} \mid \langle \text{var} \rangle$

B6)  $\langle \text{highPrecedenceOperator} \rangle \Rightarrow \text{TK\_MUL} \mid \text{TK\_DIV}$

B7)  $\langle \text{lowPrecedenceOperators} \rangle \Rightarrow \text{TK\_PLUS} \mid \text{TK\_MINUS}$

Analysis of B1: As there is only single rule available for the nonterminal  $\langle \text{arithmeticExpression} \rangle$ , we must see that is not nullable

$\text{FIRST}(\langle \text{arithmeticExpression} \rangle) = \text{FIRST}(\langle \text{term} \rangle) = \text{FIRST}(\langle \text{factor} \rangle) = \{\text{TK\_ID}, \text{TK\_NUM}, \text{TK\_RNUM}, \text{TK\_OP}\}$

There is no need to go further to check LL(1) compatibility. The grammar for  $\langle \text{arithmeticExpression} \rangle$  is LL(1) compatible.

Analysis of B2: One of the productions for the nonterminal  $\langle \text{expPrime} \rangle$  is nullable.

Let us verify  $\text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle)$  and  $\text{FOLLOW}(\langle \text{expPrime} \rangle)$  are disjoint.

$\text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle) = \text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle)$   
 $= \{\text{TK\_PLUS}, \text{TK\_MINUS}\}$  ..... using analysis of B7

and,

$\text{FOLLOW}(\langle \text{expPrime} \rangle) = \text{FOLLOW}(\langle \text{arithmeticExpression} \rangle)$  ..... using B1  
 $= \{\text{TK\_SEM}\}$  ..... using 24

This implies that

$$\begin{aligned} & \text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle) \cap \text{FOLLOW}(\langle \text{expPrime} \rangle) \\ &= \{\text{TK\_PLUS}, \text{TK\_MINUS}\} \cap \{\text{TK\_SEM}\} \\ &= \emptyset \end{aligned}$$

Therefore, B2 is also LL(1) compatible.

#### Analysis of B3:

As there is only single rule available for the nonterminal  $\langle \text{term} \rangle$ , we must see that is not nullable

$$\begin{aligned} & \text{FIRST}(\langle \text{term} \rangle) \\ &= \text{FIRST}(\langle \text{factor} \rangle) \\ &= \{\text{TK\_ID}, \text{TK\_NUM}, \text{TK\_RNUM}, \text{TK\_OP}\} \dots \dots \dots \text{using analysis of B5} \end{aligned}$$

There is no need to go further to check LL(1) compatibility. The grammar for  $\langle \text{term} \rangle$  is LL(1) compatible.

#### Analysis of B4:

One of the productions for the nonterminal  $\langle \text{termPrime} \rangle$  is nullable.

Let us verify  $\text{FIRST}(\langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle)$  and  $\text{FOLLOW}(\langle \text{termPrime} \rangle)$  are disjoint.

$$\begin{aligned} \text{FIRST}(\langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle) &= \text{FIRST}(\langle \text{highPrecedenceOperators} \rangle) \\ &= \{\text{TK\_MUL}, \text{TK\_DIV}\} \dots \dots \dots \text{using analysis of B6} \end{aligned}$$

And,

$$\begin{aligned} \text{FOLLOW}(\langle \text{termPrime} \rangle) &= \text{FOLLOW}(\langle \text{term} \rangle) \dots \dots \dots \text{using B3} \\ &= \text{FIRST}(\langle \text{expPrime} \rangle) \cup \text{FOLLOW}(\langle \text{expPrime} \rangle) \dots \dots \dots \text{using B2} \\ &= \text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle) \cup \text{FOLLOW}(\langle \text{expPrime} \rangle) \dots \dots \dots \text{using B2} \\ &= \{\text{TK\_PLUS}, \text{TK\_MINUS}\} \cup \text{FOLLOW}(\langle \text{expPrime} \rangle) \dots \dots \dots \text{using analysis of B7} \end{aligned}$$

$$\begin{aligned}
&= \{TK\_PLUS, TK\_MINUS\} \cup FOLLOW(<arithmeticExpression>) \dots \text{using B1} \\
&= \{TK\_PLUS, TK\_MINUS\} \cup \{TK\_SEM\} \dots \text{using 24} \\
&= \{TK\_PLUS, TK\_MINUS, TK\_SEM\}
\end{aligned}$$

Therefore,

$$\begin{aligned}
&FIRST(<highPrecedenceOperators><factor> <termPrime>) \cap FOLLOW(<termPrime>) \\
&= \{TK\_MUL, TK\_DIV\} \cap \{TK\_PLUS, TK\_MINUS, TK\_SEM\} \\
&= \phi
\end{aligned}$$

Hence the rules defined by B4 conform to LL(1) specifications.

Analysis of B5: There are two productions for the non terminal <factor>.

$$FIRST(TK\_OP <arithmeticExpression> TK\_CL) = \{TK\_OP\}$$

and,

$$FIRST(<var>) = \{TK\_ID, TK\_NUM, TK\_RNUM\}$$

It is now obvious that no production for <factor> is a nullable production and,

$$FIRST(TK\_OP <arithmeticExpression> TK\_CL) \cap FIRST(<var>) = \phi$$

Therefore the grammar for <factor> is LL(1) compatible.

and,

$$FIRST(<factor>) = \{TK\_ID, TK\_NUM, TK\_RNUM, TK\_OP\}$$

Analysis of B6: The two productions of the nonterminal <highPrecedenceOperators> are trivially LL(1) compatible

$$FIRST(<highPrecedenceOperators>) = \{TK\_MUL, TK\_DIV\}$$

Analysis of B7: The two productions of the nonterminal <lowPrecedenceOperators> are also trivially LL(1) compatible

$$FIRST(<lowPrecedenceOperators>) = \{TK\_PLUS, TK\_MINUS\}$$

35. <operator> ==> TK\\_PLUS | TK\\_MUL | TK\\_MINUS | TK\\_DIV

This rule is discarded now.

36.  $\langle \text{booleanExpression} \rangle \Rightarrow \text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL} \langle \text{logicalOp} \rangle \text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL}$   
 $\text{FIRST}(\text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL} \langle \text{logicalOp} \rangle \text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL}) = \{ \text{TK\_OP} \}$   
 and the production is not nullable.

37.  $\langle \text{booleanExpression} \rangle \Rightarrow \langle \text{var} \rangle \langle \text{relationalOp} \rangle \langle \text{var} \rangle$   
 $\text{FIRST}(\langle \text{var} \rangle \langle \text{relationalOp} \rangle \langle \text{var} \rangle) = \text{FIRST}(\langle \text{var} \rangle)$   
 $= \{ \text{TK\_ID}, \text{TK\_NUM}, \text{TK\_RNUM} \}$   
 and the production is not nullable.

38.  $\langle \text{booleanExpression} \rangle \Rightarrow \text{TK\_NOT} \langle \text{booleanExpression} \rangle$

Introduce the parentheses pair around the  $\langle \text{booleanExpression} \rangle$  as follows

$\langle \text{booleanExpression} \rangle \Rightarrow \text{TK\_NOT} \text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL}$  ....38-a

Considering the above rules 36-38 for analysis, we observe that none of the three productions for the nonterminal  $\langle \text{booleanExpression} \rangle$  is nullable (using First set of  $\langle \text{var} \rangle$  from 39)

and, The first sets of the right hand sides of the three productions are disjoint.

$\text{FIRST}(\text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL} \langle \text{logicalOp} \rangle \text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL})$   
 $\cap \text{FIRST}(\langle \text{var} \rangle \langle \text{relationalOp} \rangle \langle \text{var} \rangle)$   
 $\cap \text{FIRST}(\text{TK\_NOT} \text{TK\_OP} \langle \text{booleanExpression} \rangle \text{TK\_CL})$   
 $= \{ \text{TK\_OP} \} \cap \{ \text{TK\_ID}, \text{TK\_NUM}, \text{TK\_RNUM} \} \cap \{ \text{TK\_NOT} \}$   
 $= \phi$

Hence, the rules 36, 37 and 38-a for the nonterminal  $\langle \text{booleanExpression} \rangle$  are LL(1) compatible.

39.  $\langle \text{var} \rangle \Rightarrow \text{TK\_ID} \mid \text{TK\_NUM} \mid \text{TK\_RNUM}$

This rule needs modifications. The non-terminal  $\langle \text{var} \rangle$  participates in read and write statements and in arithmetic and Boolean expressions. It is expected to represent the variable identifier, the record and union variables constructed by expanding the identifier name with dot operator followed by the field name. Also  $\langle \text{var} \rangle$  represents the statically available integer and real numbers.

The rule for  $\langle \text{var} \rangle$  is modified as follows

$\langle \text{var} \rangle \Rightarrow \langle \text{singleOrRecId} \rangle \mid \text{TK\_NUM} \mid \text{TK\_RNUM}$

.....39 a.

The three rules on the RHS are LL(1) compatible as

$\text{FIRST}(\langle \text{singleOrRecId} \rangle) \cap \text{FIRST}(\text{TK\_NUM}) \cap \text{FIRST}(\text{TK\_RNUM}) = \phi$

And

$\text{FIRST}(\langle \text{var} \rangle) = \{\text{TK\_ID}, \text{TK\_NUM}, \text{TK\_RNUM}\}$

40.  $\langle \text{logicalOp} \rangle \Rightarrow \text{TK\_AND} \mid \text{TK\_OR}$

I have not defined the precedence of AND over OR or vice versa. We have the support to enclose the boolean expression in parentheses.

Trivially the productions for  $\langle \text{logicalOp} \rangle$  conform to LL(1) specifications

41.  $\langle \text{relationalOp} \rangle \Rightarrow \text{TK\_LT} \mid \text{TK\_LE} \mid \text{TK\_EQ} \mid \text{TK\_GT} \mid \text{TK\_GE} \mid \text{TK\_NE}$

All six productions above for the nonterminal  $\langle \text{relationalOp} \rangle$  are trivially LL(1) compatible and

$\text{FIRST}(\langle \text{relationalOp} \rangle) = \{\text{TK\_LT}, \text{TK\_LE}, \text{TK\_EQ}, \text{TK\_GT}, \text{TK\_GE}, \text{TK\_NE}\}$

42.  $\langle \text{returnStmt} \rangle \Rightarrow \text{TK\_RETURN} \langle \text{optionalReturn} \rangle \text{TK\_SEM}$

$\text{FIRST}(\langle \text{returnStmt} \rangle) = \{\text{TK\_RETURN}\}$

and the production is not nullable, hence the rule is LL(1) compatible.

43.  $\langle \text{optionalReturn} \rangle \Rightarrow \text{TK\_SQL} \langle \text{idList} \rangle \text{TK\_SQR} \mid \text{eps}$

$\text{FIRST}(\text{TK\_SQL} \langle \text{idList} \rangle \text{TK\_SQR}) = \{\text{TK\_SQL}\}$

and,

$\text{FOLLOW}(\langle \text{optionalReturn} \rangle) = \{\text{TK\_SEM}\}$ ..... using 42

Since both of the above sets are disjoint, the grammar for the nonterminal  $\langle \text{optionalReturn} \rangle$  is LL(1) compatible.

44.  $\langle \text{idList} \rangle \Rightarrow \text{TK\_ID} \langle \text{more\_ids} \rangle$

$\text{FIRST}(\langle \text{idList} \rangle) = \{\text{TK\_ID}\}$  and the production is not nullable, hence the rule is LL(1) compatible.

45.  $\langle \text{more\_ids} \rangle \Rightarrow \text{TK\_COMMA} \langle \text{idList} \rangle \mid \text{eps}$

$\text{FIRST}(\text{TK\_COMMA} \langle \text{idList} \rangle) = \{\text{TK\_COMMA}\}$

and,

$\text{FOLLOW}(\langle \text{more\_ids} \rangle) = \text{FOLLOW}(\langle \text{idList} \rangle)$ ..... using 44

$$= \{TK\_SQR\} \dots \text{using 43}$$

This implies that

$$\text{FIRST}(TK\_COMMA \langle idList \rangle) \cap \text{FOLLOW}(\langle more\_ids \rangle)$$

$$= \{TK\_COMMA\} \cap \{TK\_SQR\}$$

$$= \phi$$

Therefore, the rules for the nonterminal  $\langle more\_ids \rangle$  are LL(1) compatible.

46.  $\langle defintypestmt \rangle ::= TK\_DEFINETYPE \langle A \rangle TK\_RUID TK\_AS TK\_RUID$

$\text{FIRST}(\langle defintypestmt \rangle) = \{TK\_DEFINETYPE\}$  and the production is not nullable, hence the rule is LL(1) compatible.

47.  $\langle A \rangle ::= TK\_RECORD \mid TK\_UNION$

$$\text{FIRST}(TK\_RECORD) \cap \text{FIRST}(TK\_UNION)$$

$$= \{TK\_RECORD\} \cap \{TK\_UNION\} \text{ LL(1) Compatible}$$

$$= \phi \text{ ( LL(1) Compatible)}$$